Measuring Temperature with an Atari

by P. Bartram

This article shows how to use any model of Atari computer to measure temperature, using only one component, some wire and a connector.

To use a computer to measure values in the real world we need a way of converting analogue values, such as temperature, light, or resistance into digital or on/off signals which the computer's chips can deal with. Luckily Atari computers have several analogue to digital converters built in to handle the paddle controllers, two per joystick port. Paddles, which are simply variable resistors, are not the only devices that can be plugged in however, the resistance of any device can be measured and usable values are 1k ohm to 500k ohms.

IT COULDN'T BE EASIER!

Thermistors are readily available devices whose resistance varies with temperature. All that is needed is to wire up a thermistor to two pins of a suitable connector, plug it into a joystick port and with a few lines of BASIC your Atari will be measuring temperature.

We need to choose a thermistor that has a suitable resistance in the temperature range we are interested in. Type VA 1067S is a good choice as it has a resistance of 150k ohms at 25 degrees centigrade.

The parts required are available from Maplin Electronic Supplies, P.O. Box 3, Rayleigh, Essex SS6 2BR.

Thermistor VA 1067S Order code FX43W 78 pence
D-Range 9 way socket Order code RK61R 95 pence

MAKING THE RIGHT CONNECTIONS



Joystick port

1. Joystick switch - forward
2. Joystick switch - back
3. Joystick switch - left / Paddle A trigger
4. Joystick switch - right/ Paddle B trigger
5. Paddle B
6. Joystick trigger
7. +5V
8. 0V
9. Paddle A

The thermistor should be connected by suitable lengths of wire to pins 7 and 9 of the connector. These will be numbered on the socket but you need excellent eyesight or a magnifying glass to see them. The value will be read from BASIC from PADDLE(0) when plugged into joystick port 1 or PADDLE(2) in port 2. These values will vary from 1, at minimum resistance, to a maximum of 228.

A second thermistor could be connected to pins 5 and 7 and read as PADDLE(1) or PADDLE(3). In fact an Atari 400 or 800 could have up to eight analogue inputs and Atari 600XL, 800XL and 1200XL models up to four.

A DEGREE OF SUCCESS

So far so good, that's the hardware done, now for the software. Our computerised thermometer must be calibrated. Plug it into controller port 2 (then you won't need to unplug your joystick), and enter and run the following one line program:

```
10 PRINT PADDLE(2):GOTO 10
```

```
US 1000 REM ########################################
HW 1005 REM Program measures temperature
US 1010 REM using thermistor wired to
UM 1015 REM PADDLE(2) contacts.
OA 1020 REM ----------------------------
KS 1025 REM PAGE 6 MAGAZINE - ENGLAND
VB 1030 REM ########################################
ZK 1040 POKE 82,0:REM Set left margin
BI 1050 GRAPHICS 0
ZN 1060 POKE 752,1:REM Turn cursor off
UQ 1070 POKE 710,140:REM Background color
OW 1080 POKE 709,0:REM Text luminance
IX 1090 REM
LH 1100 REM Calibration data
VX 1110 LOTEMP=13.5:LOPADDLE=95
PS 1120 HITEMP=22:HIPADDLE=62
HT 1130 FACTOR=(LOPADDLE-HIPADDLE)/(LOTEM
   P-HITEMP)
VJ 1140 ? "      ATARI  Digital  Thermomet
   er"
WE 1150 ? :? "         PADDLE   TEMPERATUR
   E"
KU 1160 REM Main loop
WP 1170 T=PADDLE(2)
ZF 1180 GOSUB 5000
FM 1190 POSITION 10,5
UY 1200 PRINT T;
FF 1210 POSITION 20,5
RU 1220 PRINT INT(TEMP*10+0.5)/10
QN 1230 GOTO 1160
QZ 5000 REM ----------------------------
             )  Convert PADDLE value to (
EP 5010 REM )  temperature - degrees C.(
DK 5030 REM )  Calibration data:        (
AC 5050 REM )     temperature  paddle   (
DV 5060 REM )       13.5          95    (
UR 5070 REM )       22            62    (
PD 5090 REM )  T is paddle value        (
JI 5100 REM )  TEMP is temperature in   (
             )  deg C                   (
             ----------------------------
MZ 5120 TEMP=LOTEMP+(T-LOPADDLE)/FACTOR
AO 5130 RETURN
```

Next, put the thermistor in a cool place along with a thermometer, centigrade or fahrenheit, whichever scale you wish your program to use. After a few minutes, when readings have stabilised, note the thermometer reading and the paddle value. Give these values to variables LOTEMP and LOPADDLE. Move the thermistor and thermometer to a warm place and repeat the procedure, giving values to HITEMP and HIPADDLE. The device has a negative temperature coefficient so that when the temperature rises the resistance and paddle value fall. These four variables are used to calculate the change in resistance per degree, which is stored in the variable FACTOR. See lines 1100 to 1130 of listing 1. The above variables are used in the conversion subroutine shown in listing 1 at line 5000.

Reading the temperature in your own programs is very easy. You simply need to get the value of PADDLE(2), and change it to degrees centigrade with the conversion subroutine. The program in listing 1 does this and continuously displays both the paddle value and temperature on the screen. Hopefully you can think of a more imaginative application.

As mentioned earlier any device with a variable resistance may be used, and another very simple project is to connect up a light dependant resistor (LDR). This could then be used, for example, to detect the breaking of a beam of light falling on the LDR and then sounding an alarm, or to count objects passing in front of it. ●

FORKLIFT from page 58

```
KX 3998 REM
KQ 3999 REM *** GR. 8 ROUTINE ***
AN 4000 DIM A$(257):RESTORE 4010:FOR J=1
   TO 257:READ A:A$(J,J)=CHR$(A):NEXT J:A
   ADR=ADR(A$):RETURN
FL 4010 DATA 216,104,104,104,133,203,104,
   104,133,204,169,0,133,205,6,204,38,205
   ,6,204,38,205,6,204,38
WF 4020 DATA 205,165,204,24,101,88,133,20
   6,165,205,101,89,133,207,6,204,38,205,
   6,204,38,205,165,204,24
VX 4030 DATA 101,206,133,206,165,205,101,
   207,133,207,165,206,24,101,203,133,206
   ,141,133,6,165,207,105,0,133
YX 4040 DATA 207,141,134,6,104,133,213,10
   4,133,212,104,104,141,129,6,206,129,6,
   104,104,141,131,6
YI 4050 DATA 169,0,141,132,6,141,130,6,16
   9,0,141,128,6,172,130,6,177,212,16,5,2
   06,128,6,41,127,201,32
WF 4060 DATA 176,5,24,105,64,16,7,201,96,
   176,3,56,233,32,133,204,169,0,133,205,
   133,208,6,204,38
HE 4070 DATA 205,6,204,38,205,6,204,38,20
   5,165,205,24,109,244,2,133,205,164,208
   ,177,204,77,128,6,172
UU 4075 DATA 132,6,145,206
VC 4080 DATA 230,208,165,208,201,8,240,15
   ,165,206,24,105,40,133,206,144,227,230
   ,207,208,223
UB 4090 DATA 144,160,238,132,6,238,130,6,
   206,129,6,48,43,173,132,6,205,131,6,20
   8,22,169,0,141,132,6,24,173,133,6
BO 4100 DATA 105,64,141,133,6,173,134,6,1
   05,1,141,134,6,173,133,6,133,206,173,1
   34,6,133,207,24,144,200,96
```

□